

NAG Library Routine Document

G05SCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SCF generates a vector of pseudorandom numbers from a Cauchy distribution with median a and semi-interquartile range b .

2 Specification

```
SUBROUTINE G05SCF(N, XMED, SEMIQR, STATE, X, IFAIL)
INTEGER          N, STATE(lstate), IFAIL
double precision XMED, SEMIQR, X(N)
```

3 Description

The distribution has PDF (probability density function)

$$f(x) = \frac{1}{\pi b \left(1 + \left(\frac{x-a}{b}\right)^2\right)}.$$

G05SCF returns the value

$$a + b \frac{2y_1 - 1}{y_2},$$

where y_1 and y_2 are a pair of consecutive pseudorandom numbers from a uniform distribution over $(0, 1)$, such that

$$(2y_1 - 1)^2 + y_2^2 \leq 1.$$

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SCF.

4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin
 Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

5 Parameters

- | | | |
|----|--|--------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of pseudorandom numbers to be generated. | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | XMED – <i>double precision</i> | <i>Input</i> |
| | <i>On entry:</i> a , the median of the distribution. | |
| 3: | SEMIQR – <i>double precision</i> | <i>Input</i> |
| | <i>On entry:</i> b , the semi-interquartile range of the distribution. | |
| | <i>Constraint:</i> $SEMIQR \geq 0.0$. | |

- 4: STATE(*lstate*) – INTEGER array *Communication Array*
Note: *lstate* = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 5: X(N) – **double precision** array *Output*
On exit: the *n* pseudorandom numbers from the specified Cauchy distribution.
- 6: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or –1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 3

On entry, SEMIQR < 0.0.

IFAIL = 4

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints the first five pseudorandom real numbers from a Cauchy distribution with median 1.0 and semi-interquartile range 2.0, generated by a single call to G05SCF, after initialization by G05KFF.

9.1 Program Text

```

*      G05SCF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, N
      PARAMETER        (MSTATE=633,MSEED=1,N=5)
*      .. Local Scalars ..
      DOUBLE PRECISION SEMIQR, XMED
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*      .. External Subroutines ..
      EXTERNAL         G05KFF, G05SCF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05SCF Example Program Results'
      WRITE (NOUT,*)
*      Set the distribution parameters XMED and SEMIQR
      XMED = 1.0D0
      SEMIQR = 2.0D0
*      Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*      GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*      Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*      Generate the variates
      IFAIL = 1
      CALL G05SCF(N,XMED,SEMIQR,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*      Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999  FORMAT (1X,F10.4)
99998  FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997  FORMAT (1X,' ** G05SCF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05SCF Example Program Results

```

6.1229
2.2328
-2.2118
0.4118
0.9892

```